UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

A computer program for borehole compensation of dual-detector density well logs

by

James H. Scott

Federal Center, Denver, Colorado 80225

Open-File Report 78-515

A computer program for borehole compensation of dual-detector density well logs

by

James H. Scott

U.S. Geological Survey, Denver, Colorado 80225

Abstract

The computer program described in this report was developed for applying a borehole-rugosity and mudcake compensation algorithm to dual-density logs using the following information: the water level in the drill hole, hole diameter (from a caliper log if available, or the nominal drill diameter if not), and the two gamma-ray count rate logs from the near and far detectors of the density probe. The equations that represent the compensation algorithm and the calibration of the two detectors (for converting countrate or density) were derived specifically for a probe manufactured by Comprobe Inc.[1/] (5.4 cm O.D. dual-density-caliper); they are not applicable to other probes. However, equivalent calibration and compensation equations can be empirically determined for any other similar two-detector density probes and substituted in the computer program listed in this report.

1/ Use of brand names in this report does not necessarily constitute endorsement by the U.S. Geological Survey.

Introduction

The computer program described and listed in this report is part of a generalized log interpretation program developed and used by the USGS on the Honeywell Multics computing system. This system requires a somewhat different format for fortran statements than is used in standard fortran; for example, lower-case ASCII characters are used, and continuation-card numbers are not needed in column 6. However, the Multics fortran format can be converted to standard fortran format for use with other computers. The program described and listed in this report is based on a density log interpretation technique discussed by Scott (1977) and developed originally by Wahl and others (1964).

Program Description

The computational program segment listed in this report must be preceded by an input routine for reading both the far-detector and the near-detector count rate logs. The far-detector log is digitized at constant-depth intervals (usually 0.5 ft) and is stored in array xlog(i,1) with corresponding depths stored in array zlog(i,1). The near-detector count rate log is digitized at the same constant depth intervals and is stored in array xlog(i,2) with corresponding depths stored in array zlog(i,2). It is assumed that dead-time corrections have been previously applied to count rate data stored in these arrays. The digitized caliper log, if available, is stored in xlog(i,3) with depths stored in zlog(i,3). The depth of the water level is stored as "wl", and the nominal diameter as "diam". The index of the last (deepest) data point in arrays xlog(i,j) and zlog(i,j) is stored in array nm(j) where j=1 represents the far-detector log, j=2 represents the near-detector log, and j=3 represents the caliper log.

The program begins by alining the depth of the near-detector and caliper logs with the depth of the far-detector log at the i^{th} data point using the align subrouting also listed in this report. Then, depending on whether the i^{th} data point is above or below the water level, near- and far-detector count rates are converted to apparent density, after which the compensation formula is applied to compute a compensated density value for dry or water-filled boreholes. The computed value of compensated density is stored back in array xlog(i,1). If only the far-detector count rate is available (because the near-detector log is missing at the i^{th} depth point), the compensation formula is skipped and the apparent density is computed for the far-detector log only. This result is stored in array xlog(i,1). At the end of the computation the output density and depth values are stored in xlog(i,1), zlog(i,1) for subsequent printing or plotting using output routines of the user's choice.

EXAMPLE

An example of results obtained by use of the program is given below along with input data that can be used as a test case. In the example water level (wl) is 25.0 ft and nominal hole diameter (diam) is 5.0 inches.

	Input data						Output data	
	Far detector log		Near detector log		Caliper		Compensated density log	
	Depth(ft)	count rate(cps)	Depth(ft)	Count rate(cps)	Depth(ft)	Diam.(in)	Depth(ft)	Density(gm/cm^3)
Array	zlog(i,1)	xlog(i,1)	zlog(i,2)	xlog(i,2)	zlog(i,3)	xlog(i,3)	zlog(i,1)	xlog(i,1)
	20.0	40	20.5	400	16.0	4.88	20.0	2.669
	20.5	52	21.0	405	16.5	5.02	20.5	2.617
	21.0	58	21.5	410	17.0	5.01	21.0	2.524
	21.5	63	22.0	417	17.5	4.89	21.5	2.467
	22.0	70	22.5	427	18.0	4.79	22.0	2.403
	22.5	76	23.0	431	18.5	4.73	22.5	2.358
	23.0	81	23.5	437	19.0	4.70	23.0	2.323
	23.5	89	24.0	457	19.5	4.65	23.5	2.276
	24.5	99	25.0	462	20.5	4.60	24.5	2.213
	25.0	108	25.5	468	21.0	4.59	25.0	2.169
	25.5	115	26.0	479	21.5	4.60	25.5	2.102
	26.0	137	26.5	495	22.0	4.60	26.0	2.017
	26.5	143	27.0	498	22.5	4.61	26.5	1.991
	27.0	158	27.5	505	23.0	4.63	27.0	1.941
	27.5	162	28.0	514	23.5	4.68	27.5	1.927
	28.0	175	28.5	527	24.0	4.72	28.0	1.887
	28.5	189	29.0	538	24.5	4.80	28.5	1.849
	29.0	193	29.5	542	25.0	4.85	29.0	1.845
	29.5	205	30.0	543	25.5	4.91	29.5	1.813
	30.0	212	30.5	543	26.0	4.97	30.0	1.794

Program listing

```
c
c ALIGN DEPTH OF NEAR DETECTOR AND CALIPER LOGS WITH DEPTH OF FAR DETECTOR LOG
c
          nfar=nm(1)
          xdiam=diam
          do 1040 i=1,nfar
          zfar=zlog(i,1)
          xfar=xlog(i,1)
          if(nlog.eq.1) go to 1010
          call align(i,xalign)
          xnear=xalign(2)
          if(nlog.eq.3.and.xalign(3).ne.0.) xdiam=xalign(3)
c
c COMPUTE APPARENT AND COMPENSATED DENSITIES
c
1010      if(zfar.gt.wl) go to 1020
c
c ABOVE WATER LEVEL
c
          xfar=alog10(xfar*corf*(1.159-.053*xdiam))
          xfar=((xfar*cfa(4)+cfa(3))*xfar+cfa(2))*xfar+cfa(1)
          if(xnear.eq.0..or.nlog.eq.1) go to 1030
          xnear=xnear*corn*(1.012-.004*xdiam)
          xnear=((xnear*cna(4)+cna(3))*xnear+cna(2))*xnear+cna(1)
          xn2=xnear*xnear
          xlog(i,1)=((.40534237*xn2+1.1424375)*xnear-5.3080027)*xnear
           +((2.2139486*xfar-9.9475336)*xfar+14.719678)*xfar
           +xnear*xfar*(2.2207969-.67623508*xn2)-3.8736607
          go to 1040
c
c BELOW WATER LEVEL
c
1020      xfar=alog10(xfar*corf*(1.081-.027*xdiam))
          xfar=((xfar*cfw(4)+cfw(3))*xfar+cfw(2))*xfar+cfw(1)
          if(xnear.eq.0..or.nlog.eq.1) go to 1030
          xnear=xnear*corn*(1.009-.003*xdiam)
          xnear=((xnear*cnw(4)+cnw(3))*xnear+cnw(2))*xnear+cnw(1)
          xlog(i,1)=(1.2431213*xnear-7.6101682)*xnear
           +(((-.06728998*xfar+2.5029254)*xfar-6.2356905)*xfar+10.941696)*xfar
           +(5.3452706-1.9531422*xfar)*xnear*xfar-.99360996
          go to 1040
c
c ONLY FAR DETECTOR DATA ARE AVAILABLE (ifar=1)
c
1030      xlog(i,1)=xfar
1040      continue
c
          do 1050 i=1,nfar
          if(xlog(i,1).le.3.0.and.xlog(i,1).ge.1.0) go to 1060
1050      xlog(i,1)=0.
1060      continue
```

```
      subroutine align(i,x)
c
c This subroute aligns depths of 1 or 2 logs with a reference log stored
c in arrays xlog(i,1),zlog(i,1).  The logs to be aligned are stored in
c arrays xlog(i,2),zlog(i,2) if one log is to be aligned, and in arrays
c xlog(i,2),zlog(i,2) and xlog(i,3),zlog(i,3) if two logs are to be aligned.
c
c   nlog = number of logs to be aligned
c      i = depth index of reference log at alignment depth zlog(i,1)
c      x = array of x-values aligned with xlog(i,1):  x(1)=xlog(i,1)
c
      common xlog,zlog,nm,nlog
c
      dimension xlog(5000,3),zlog(5000,3),x(3),int(3),nm(3)
c
      x(1)=xlog(i,1)
      zref=zlog(i,1)
      if(i.gt.1) go to 20
      do 10 k=2,nlog
10    int(k)=ifix((zlog(1,1)-zlog(1,k))/(zlog(2,1)-zlog(1,1)))
c
20    do 70 k=2,nlog
      x(k)=0.
      j=i+int(k)
      if(j.le.0.or.j.gt.nm(k)) go to 70
      if(zlog(j,k).eq.zref) go to 60
      if(zlog(j,k).lt.zref) go to 40
30    j=j-1
      if(j.le.0) go to 70
      if(zlog(j,k).gt.zref) go to 30
      if(zlog(j,k).eq.zref) go to 60
      j1=j
      j2=j+1
      go to 50
c
40    j=j+1
      if(j.gt.nm(k)) go to 70
      if(zlog(j,k).lt.zref) go to 40
      if(zlog(j,k).eq.zref) go to 60
      j1=j-1
      j2=j
50    x(k)=((zref-zlog(j1,k))*(xlog(j2,k)-xlog(j1,k)))/
      (zlog(j2,k)-zlog(j1,k))+xlog(j1,k)
      go to 70
c
60    x(k)=xlog(j,k)
c
70    continue
      return
      end
```

References

Scott, James H., 1977, Borehole compensation algorithms for a small-diameter, dual-detector density well-logging probe: Transactions SPWLA 18th Ann. Logging Symposium, p. 51-517.

Wahl, J. S., Tittman, J., Johnson, C. W., And Alger, R. P., 1964, The dual spacing formation density log: Jour. of Pet. Tech., v. 16, p. 1411-1416.